
G13 – Time Series Analysis

G13AHF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

G13AHF produces forecasts of a time series, given a time series model which has already been fitted to
the time series using G13AEF or G13AFF. The original observations are not required, since G13AHF
uses as input either the original state set produced by G13AEF or G13AFF or the state set updated by
a series of new observations using G13AGF. Standard errors of the forecasts are also provided.

2 Specification

SUBROUTINE G13AHF(ST, NST, MR, PAR, NPAR, C, RMS, NFV, FVA, FSD,
1 WA, NWA, IFAIL)
INTEGER NST, MR(7), NPAR, NFV, NWA, IFAIL
real ST(NST), PAR(NPAR), C, RMS, FVA(NFV), FSD(NFV),
1 WA(NWA)

3 Description

The original time series is xt, for t = 1, 2, . . . , n and parameters have been fitted to the model of this
time series using G13AEF or G13AFF.

Forecasts of xt, for t = n+ 1, n+ 2, . . . , n+ L, are calculated in five stages, as follows:

(i) set at = 0 for t = N +1, N +2, . . . , N +L, where N = n− d− (D× s) is the number of differenced
values in the series,

(ii) calculate the values of et for t = N + 1, N + 2, . . . , N + L and et = φ1 × et−1 + . . . + φp × et−p +
at − θ1 × at−1 − . . .− θq × at−q,

(iii) calculate the values of wt for t = N + 1, N + 2, . . . , N + L, where wt = Φ1 × wt−s + . . . + ΦP ×
wt−s×P + et − Θ1 × et−s − . . . − ΘQ × et−s×Q and wt for t ≤ N are the first s × P values in the
state set, corrected for the constant,

(iv) add the constant term c to give the differenced series∇d∇D
s xt = wt+c for t = N+1, N+2, . . . , N+L,

(v) the differencing operations are reversed to reconstitute xt for t = n+ 1, n+ 2,+L.

The standard errors of these forecasts are given by st = {V × (ψ2
0 + ψ2

1 + . . . + ψ2
t−n−1)}1/2 for

t = n + 1, n + 2, . . . , n + L, where ψ0 = 1, V is the residual variance of at and ψj is the coefficient
expressing the dependence of xt on at−j .

To calculate ψj for j = 1, 2, . . . , (L− 1) the following device is used.

A copy of the state set is initialised to zero throughout and the calculations outlined above for the
construction of forecasts are carried out with the settings aN+1 = 1, and at = 0 for t = N + 2, N +
3, . . . , N + L.

The resulting quantities corresponding to the sequence xN+1, xN+2, . . . , xN+L are precisely 1,
ψ1, ψ2, . . . , ψL−1.

The supplied time series model is used throughout these calculations, with the exception that the constant
term c is taken to be zero.

4 References

None.

5 Parameters

1: ST(NST) — real array Input

On entry: the state set derived from G13AEF or G13AFF originally, or as modified using earlier
calls of G13AGF.

2: NST — INTEGER Input

On entry: the number of values in the state set array ST.

Constraint: NST = P × s+D× s+ d+ q+max(p,Q× s). (As returned by G13AEF or G13AFF).

3: MR(7) — INTEGER array Input

On entry: the orders vector (p, d, q, P,D,Q, s) of the ARIMA model, in the usual notation.

Constraints:

p, d, q, P,D,Q, s ≥ 0,
p+ q + P +Q > 0,
s �= 1,
if s = 0, then P +D +Q = 0,
if s > 1, then P +D +Q > 0

4: PAR(NPAR) — real array Input

On entry: the estimates of the p values of the φ parameters, the q values of the θ parameters, the
P values of the Φ parameters and the Q values of the Θ parameters which specify the model and
which were output originally by G13AEF or G13AFF.

5: NPAR — INTEGER Input

On entry: the number of φ, θ, Φ and Θ parameters in the model.

Constraint: NPAR = p+ q + P +Q.

6: C — real Input

On entry: the value of the model constant, c. This will have been output by G13AEF or G13AFF.

7: RMS — real Input

On entry: the residual variance, V, associated with the model. If G13AFF was used to estimate the
model, RMS should be set to S/NDF, where S and NDF were output by G13AFF. If G13AEF was
used to estimate the model, RMS should be set to S/ICOUNT(5), where S and ICOUNT(5) were
output by G13AEF.

Constraint: RMS ≥ 0.0.

8: NFV — INTEGER Input

On entry: the required number of forecasts, L.

Constraint: NFV > 0.

9: FVA(NFV) — real array Output

On exit: NFV forecast values relating to the original undifferenced series.

10: FSD(NFV) — real array Output

On exit: the standard errors associated with each of the NFV forecast values in FVA.

11: WA(NWA) — real array Workspace

12: NWA — INTEGER Input

On entry: the dimension of the array WA as declared in the (sub)program from which G13AHF is
called.

Constraint: NWA ≥ (4×NPAR + 3×NST).

13: IFAIL — INTEGER Input/Output

On entry: IFAIL must be set to 0, −1 or 1. For users not familiar with this parameter (described
in Chapter P01) the recommended value is 0.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, NPAR �= p+ q + P +Q,

or the orders vector MR is invalid (check the constraints given in Section 5).

IFAIL = 2

On entry, NST �= P × s+D × s+ d+ q +max(Q× s, p).

IFAIL = 3

On entry, NFV ≤ 0.

IFAIL = 4

On entry, NWA < 4×NPAR+ 3×NST.

IFAIL = 5

On entry, RMS < 0.0.

7 Accuracy

The computations are believed to be stable.

8 Further Comments

The time taken by the routine is approximately proportional to NFV×NPAR.

9 Example

The following program is based on the data derived in the example used to illustrate G13AGF.

These consist of a set of orders indicating that there are two moving average parameters (one non-seasonal,
and one seasonal with periodicity 12).

The model constant is zero.

The state set contains 26 values.

In addition the residual mean-square derived when the model was originally fitted is given.

Twelve forecasts and their associated errors are obtained.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* G13AHF Example Program Text
* Mark 14 Revised. NAG Copyright 1989.
* .. Parameters ..

INTEGER NPMAX, NSTMAX, NWAMAX, NFVMAX
PARAMETER (NPMAX=10,NSTMAX=40,NWAMAX=120,NFVMAX=25)
INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)

* .. Local Scalars ..
real C, RMS
INTEGER I, IFAIL, NFV, NPAR, NST, NWA

* .. Local Arrays ..
real FSD(NFVMAX), FVA(NFVMAX), PAR(NPMAX), ST(NSTMAX),

+ WA(NWAMAX)
INTEGER MR(7)

* .. External Subroutines ..
EXTERNAL G13AHF

* .. Intrinsic Functions ..
INTRINSIC MAX

* .. Executable Statements ..
WRITE (NOUT,*) ’G13AHF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) NFV
READ (NIN,*) (MR(I),I=1,7)
NPAR = MR(1) + MR(3) + MR(4) + MR(6)
NST = MR(4)*MR(7) + MR(5)*MR(7) + MR(2) + MR(3) + MAX(MR(1),MR(6)

+ *MR(7))
NWA = 4*NPAR + 3*NST
IF (NFV.GT.0 .AND. NFV.LE.NFVMAX .AND. NPAR.GT.0 .AND. NPAR.LE.

+ NPMAX .AND. NST.GT.0 .AND. NST.LE.NSTMAX) THEN
READ (NIN,*) (PAR(I),I=1,NPAR), C
READ (NIN,*) (ST(I),I=1,NST)
READ (NIN,*) RMS
IFAIL = 0

*
CALL G13AHF(ST,NST,MR,PAR,NPAR,C,RMS,NFV,FVA,FSD,WA,NWA,IFAIL)

*
WRITE (NOUT,*)
WRITE (NOUT,99998) ’The required’, NFV,

+ ’ forecast values are as follows’
WRITE (NOUT,99999) (FVA(I),I=1,NFV)
WRITE (NOUT,*)
WRITE (NOUT,*)

+ ’The standard deviations corresponding to the forecasts are’
WRITE (NOUT,99999) (FSD(I),I=1,NFV)

END IF
STOP

*
99999 FORMAT (1X,8F8.4)
99998 FORMAT (1X,A,I3,A)

END

9.2 Program Data

G13AHF Example Program Data
12
0 1 1 0 1 1 12
0.3270 0.6262 0.0000
0.0660 -0.0513 0.1715 -0.0249 0.0588 0.1167 0.1493 0.0199

-0.1884 -0.1289 -0.1172 0.1122 6.0039 0.0443 -0.0070 0.0252
0.0020 0.0353 -0.0460 0.0374 0.0151 -0.0237 0.0031 0.0188
0.0066 0.0125
0.0014

9.3 Program Results

G13AHF Example Program Results

The required 12 forecast values are as follows
6.0381 5.9912 6.1469 6.1207 6.1574 6.3029 6.4288 6.4392
6.2657 6.1348 6.0059 6.1139

The standard deviations corresponding to the forecasts are
0.0374 0.0451 0.0517 0.0575 0.0627 0.0676 0.0721 0.0764
0.0805 0.0843 0.0880 0.0915

